DYNAMIC ANALYSIS OF A CRANK-ROCKER MECHANISM

by

ULRICH SIELAFF

B.S., State University of Iowa, 1965

A MASTER'S REPORT

submitted in partial fulfillment of the

requirements for the degree

MASTER OF SCIENCE

Department of Mechanical Engineering

KANSAS STATE UNIVERSITY
Manhattan, Kansas

1966

Approved by:

Hugh S. Walker
Major Advisor

TABLE OF CONTENTS

DYNAMIC ANALYSIS OF A CRANK-ROCKER MECHANISM

by

ULRICH SIELAFF

B.S., State University of Iowa, 1965

AN ABSTRACT OF A MASTER'S REPORT

submitted in partial fulfillment of the

requirements for the degree

MASTER OF SCIENCE

Department of Mechanical Engineering

KANSAS STATE UNIVERSITY
Manhattan, Kansas

1966

This report presents a computer program written in FORTRAN which may be used to dynamically analyze a general crank-rocker mechanism. Although analysis in one position is not a new problem, little work has been done to analyze a mechanism over a complete range of angular positions. With this program, velocities, accelerations, and forces in a linkage may be found for any range of positions of the input crank.

NOMENCLATURE

A_L, B_L, C_L	normalized length of links (dimensionless)
ψ, λ, β	defined angles (degrees)
a, b, c, d	length of links (inches)
θ_z	angular position of link z (degrees) + CCW, - CW
$\dot{\theta}_z$	angular velocity of link z (radians/sec) + CCW, - CW
$\ddot{\theta}_z$	angular acceleration of link z (radians/sec^2) + CCW, - CW
i	the imaginary number $\sqrt{-1}$
r_z	length of a vector along link z (inches)
$\bar{R}_c$	position vector to a point C (inches)
$\bar{v}_c$	velocity vector of a point C (in/sec)
$\bar{a}_c$	acceleration vector of a point C (in/sec^2)
r_c^x, r_c^y	components of position (inches)
v_c^x, v_c^y	components of velocity (in/sec)
a_c^x, a_c^y	components of acceleration (in/sec^2)
m_z	mass of link z (lbs-sec^2/ft)
I_z	mass moment of inertia (in-lbs-sec^2)
$\bar{T}_z$	torque vector on link z (in-lbs)
$\bar{F}_{uv}$	force vector of the uth link on the vth link (lbs)
δ	angular position of center of mass (degrees)

INTRODUCTION

The kinematic and dynamic analysis of four-bar mechanisms is a common problem in the study of mechanical linkages and other machine elements. Even for one position of a mechanism, the solution to such a problem is extremely tedious and time consuming. Usually, a graphical procedure involving velocity, acceleration, and force polygons is used to completely analyze a four-bar mechanism. A great number of calculations are necessary - many of which are interdependent. That is, the results obtained graphically from a velocity vector polygon are used to calculate quantities which must be used in an acceleration polygon. These, in turn, are applied to force polygons.

In such an analysis, errors may be introduced in three ways: 1. measurement of magnitude, 2. measurement of angular direction, 3. slide rule round-off error. These problems often become acute when a particular vector is extremely small with respect to another. Accurately scaling these relatively small quantities is often impossible. As a result, the solution may be no more than approximation.

It must be remembered that a graphical analysis is only good for one position of a mechanism. For each position a new set of vector polygons must be constructed in order to find the velocities, accelerations, and forces. Furthermore, the analysis must be repeated when the angular velocity of the input link is changed. The resulting accelerations and forces cannot be simply "scaled up" since these quantities depend on the squares of the angular velocities.

In this report a computer program is described which analyzes at least one type of four-bar mechanism. The need for such work was emphasized at the International Conference on Mechanisms held at Yale University in 1961 [1].* Most of the programs available at that time dealt only with the kinematics of a linkage. To date, there are not many references available that show that the dynamics of such a problem have been fully investigated. This paper deals with a computer solution to such a problem.

An effort has been made to keep the program presented here as general as possible. The masses, moments of inertia, lengths, positions of center of mass, etc., are all left as variable input quantities. With this program, a mechanism may be analyzed over any range and with as many positions of the input crank as are necessary. The output quantities are also easily controlled, including velocity, acceleration of any point, pin forces, shaking forces, and driving torque.

In order to apply this program successfully to a design problem, it is expected that the reader has a basic understanding of the FORTRAN programming language. This program was written for the IBM 1410, but could easily be modified for any computer using FORTRAN.

*Numbers in brackets designate references at end of report.

THE FOUR-BAR MECHANISM



Figure 1. The general four-bar with lengths normalized with respect to link 2.

A general four-bar mechanism is shown in Figure 1. One of the links is fixed in a stationary frame of reference and the remaining three are free to oscillate or rotate. For a given position of any one of the three movable links, the positions of the remaining two are specified. Furthermore, any link may be "extended" to any point in the plane and the position of such a point, i.e. x, is also specified.

Work has been done in the classification of four-bar mechanisms. By specifying the lengths of the members, and the fixed link, it is possible to predict what sort of oscillation or rotation may exist.

Four-bar mechanisms are usually classified as follows: [2]

class a: one crank (a link pinned to ground at one end) can rotate 360° while the other crank can only oscillate.

class b: both cranks may rotate through 360°.

class c: both cranks can oscillate but neither can make a complete revolution.

Geometric considerations for class "b" and "c" mechanisms can often become very complicated. For this reason only the class "a" mechanism is considered in this paper. This configuration is commonly called the "crank-rocker" mechanism.

Link 1 will always be considered the fixed link, and link 2 the driving link (the one which may rotate 360°). If the lengths are normalized with respect to link 2, the following conditions must hold in order that the mechanism is a crank-rocker:

1. Driving link must be the shortest link.
2. $C_L < (A_L + B_L - 1)$.
3. $C_L > (|A_L - B_L| + 1)$.

These conditions insure that the rocker does not drop below the fixed link. When this happens the mechanism becomes class "c" and an unpredictable configuration may be achieved - the rocker may return by its normal path, or could continue on around. The first part of the program makes these tests to insure that the lengths which are input will result in a crank-rocker mechanism. If these tests are not satisfied, the program will print out this information and stop.

GEOMETRY

In order to be programmed, a solution to this problem must be completely analytical. That is, no intuitive decisions as to the directions of vectors or angular velocities can be made by the computer. Completely specifying the geometry is the first step in an analytical solution.



Case I

Case II

Figure 2. Crank-rocker in 2 positions

For any given θ_2, θ_3 and θ_4 must be known. They may be found by constructing the line O_BA. This length ℓ may be determined from the law of cosines, since θ_2, a and d are known. Next, the angles ψ, λ, β are defined. The cosine of each of these angles may be determined from the law of cosines since ℓ is now known.

Examination of the mechanism reveals that there are two configurations which must be considered when using ψ, λ, and β to determine θ_3 and θ_4. Case I, when θ_2 is between 0° and 180°

and Case II when θ_2 is between 180° and 360°.

For Case I

$$\theta_3 = \psi - \beta$$
$$\theta_4 = 180° - (\lambda + \beta) .$$

For Case II

$$\theta_3 = \psi + \beta$$
$$\theta_4 = 180° - (\lambda - \beta).$$

A somewhat similar method for determining the geometry appears in an article by George H. Martin in "Machine Design Magazine" [3] . This method is reprinted in <u>Theory of Machines</u>, by Joseph E. Shigley [4]. Although the basic geometric considerations are correct, for some positions of a mechanism this derivation yields incorrect results. The error seems to be in the author's definition of Cases I and II. For mechanisms discussed in this report, the relative position of link 3, the criterion used by Mr. Martin, is not the correct criterion for determining these cases.

At least for a crank-rocker it was found that only the angular position of link 2 determines at what point the sums and differences of ψ , λ , and β must be changed to find θ_3 and θ_4 . This point is $\theta_2 = 180°$. A complete derivation is given in Appendix A.

THE KINEMATIC PROBLEM

A convenient method of representing the positions of points on a mechanism is by the use of vectors in the complex plane. The location "P" is specified by the complex number (a + bi). An equivalent statement which specifies this position is $r(\cos\theta + i\sin\theta)$. Noting that $(\cos\theta + i\sin\theta) = e^{i\theta}$, this reduces to $re^{i\theta}$.



Figure 3. Complex plane representation.

The magnitude r can be related to the length of a link and the angle θ to its angular position measured from the horizontal. This This is done in Figure 4.



Figure 4. Vectors related to a linkage.

A vector summation equation can now be written for the four vectors:

$$r_1 e^{i\theta_1} + r_4 e^{i\theta_4} = r_2 e^{i\theta_2} + r_3 e^{i\theta_3}, \tag{1}$$

since θ_1, and the r's are constant, differentiation of this expression yields:

$$i r_4 \dot{\theta}_4 e^{i\theta_4} = i r_2 \dot{\theta}_2 e^{i\theta_2} + i r_3 \dot{\theta}_3 e^{i\theta_3}. \tag{2}$$

Notice that $\dot{\theta}_2$, $\dot{\theta}_3$ and $\dot{\theta}_4$ are angular velocity terms ($\dot{\theta}_2$ is assumed known, since this is the angular velocity of the driver link). Separating the real and imaginary parts of equation (2), two equations in the unknowns $\dot{\theta}_3$ and $\dot{\theta}_4$ are obtained. These are solved:

$$\dot{\theta}_3 = \frac{r_2}{r_3} \dot{\theta}_2 \frac{\sin(\theta_4 - \theta_2)}{\sin(\theta_3 - \theta_4)}, \tag{3}$$

$$\dot{\theta}_4 = \frac{r_2}{r_4} \dot{\theta}_2 \frac{\sin(\theta_3 - \theta_2)}{\sin(\theta_3 - \theta_4)}. \tag{4}$$

Similarly, differentiating equation (2) yields:

$$i r_4 \ddot{\theta}_4 e^{i\theta_4} - r_4 \dot{\theta}_4^2 e^{i\theta_4} = i r_2 \ddot{\theta}_2 e^{i\theta_2} - r_2 \dot{\theta}_2^2 e^{i\theta_2} + i r_3 \ddot{\theta}_3 e^{i\theta_3} - r_3 \dot{\theta}_3^2 e^{i\theta_3}. \tag{5}$$

Separating the real and imaginary parts of this equation yields two equations in the unknowns $\ddot{\theta}_3$ and $\ddot{\theta}_4$. These are solved:

$$\ddot{\theta}_3 = \frac{-r_2 \ddot{\theta}_2 \sin(\theta_4 - \theta_2) + r_2 \dot{\theta}_2^2 \cos(\theta_4 - \theta_2) + r_3 \dot{\theta}_3^2 \cos(\theta_4 - \theta_3) - r_4 \dot{\theta}_4^2}{r_3 \sin(\theta_4 - \theta_3)}, \tag{6}$$

$$\ddot{\theta}_4 = \frac{r_2 \ddot{\theta}_2 \sin(\theta_3 - \theta_2) - r_2 \dot{\theta}_2^2 \cos(\theta_3 - \theta_2) + r_4 \dot{\theta}_4^2 \cos(\theta_3 - \theta_4) - r_3 \dot{\theta}_3^2}{r_4 \sin(\theta_3 - \theta_4)}. \tag{7}$$

A complete derivation of these relations is given in Appendix B.



Figure 5. Vectors to any point on a linkage.

An expression for the position of any point "C" (Figure 5) can now be written. In this case, "C" is fixed on link 3.

$$\bar{R}_c = r_2 e^{i\theta_2} + r_c e^{i(\theta_3+\delta)}. \tag{8}$$

Since velocity of C is $\dfrac{d\bar{R}_c}{dt}$,

$$V_c = i r_2 \dot{\theta}_2 e^{i\theta_2} + i r_c \dot{\theta}_3 e^{i(\theta_3+\delta)}. \tag{9}$$

Separating real and imaginary parts yields horizontal and vertical components of V_c :

$$V_c^x = -r_2 \dot{\theta}_2 \sin\theta_2 - r_c \theta_3 \sin(\theta_3+\delta), \tag{10}$$

$$V_c^y = r_2 \dot{\theta}_2 \cos\theta_2 + r_2 \dot{\theta}_3 \cos(\theta_3+\delta). \tag{11}$$

Since acceleration of C is $\dfrac{d^2\bar{R}_c}{dt^2}$,

$$\bar{a}_c = -r_2 \dot{\theta}_2^2 e^{i\theta_2} + i r_2 \ddot{\theta}_2 e^{i\theta_2} - r_c \dot{\theta}_3^2 e^{i(\theta_3+\delta)} + i r_c \ddot{\theta}_3 e^{i(\theta_3+\delta)}. \tag{12}$$

Which, when separated yields:

$$a_c^x = -r_2\dot{\theta}_2^2\cos\theta_2 - r_2\ddot{\theta}_2\sin\theta_2 - r_c\dot{\theta}_3^2\cos(\theta_3+\delta) - r_c\ddot{\theta}_3\sin(\theta_3+\delta) \quad (13)$$

$$a_c^y = -r_2\dot{\theta}_2^2\sin\theta_2 + r_2\ddot{\theta}_2\cos\theta_2 - r_c\dot{\theta}_3^2\cos(\theta_3+\delta) + r_c\ddot{\theta}_3\cos(\theta_3+\delta). \quad (14)$$

It is interesting to note that in the complex forms of V_c and a_c, each term corresponds to a term which would be used in the graphical vector analysis. For example, $i r_2\dot{\theta}_2 e^{i\theta_2}$ is equivalent to the absolute velocity of point A both in magnitude and direction.

Since

$$\begin{aligned} ie^{i\theta} &= i(\cos\theta + i\sin\theta) \\ &= i\cos\theta - \sin\theta \\ &= \cos(\theta+90) + i\sin(\theta+90) \\ &= e^{i(\theta+90)}. \end{aligned}$$

thus multiplication by i rotates a vector 90°. Therefore:

$$i r_2\dot{\theta}_2 e^{i\theta_2} = r_2\dot{\theta}_2 e^{i(\theta_2+90)}.$$

and this, in fact, represents a vector of magnitude $r_2\dot{\theta}_2$ tangential to the arc which the point A describes. Similar arguments apply to all the terms in the complex expressions. [5,6,7]

THE DYNAMIC PROBLEM

Up to this point, no simplifying assumptions have been used. In the force analysis, however, it must be assumed that the links behave as perfectly rigid bodies. This assumption allows the mass of each member to be concentrated at a single point. Although in reality these members would indeed be elastic, the problem of considering deflection due to varying inertia loads would be extremely complex. Presumably, these deflections would be small for most cases. It should be noted, however, that this program would be helpful in such a consideration since it may easily be modified to output the inertia forces on incremented masses along any link.

Basically two principles are applied. They are:

1. the D'Alembert principle, which in effect reduces a dynamic problem to one of statics.
2. the principle of superposition, which makes it possible to analyze one link at a time and then combine the results in the end.

Knowing the acceleration of the center of mass of each link and its angular acceleration, the inertia force and torque acting on each of these members may be calculated. The D'Alembert principle states that at any instant this may be treated as a problem with no motion, with the inertia terms treated as static forces. Thus

$$\Sigma \bar{F} - m\bar{a} = 0,$$
$$\Sigma \bar{T} - I\ddot{\theta} = 0.$$

In the force analysis each member is individually considered to have mass, while the others are assumed weightless. First, consider 2 and 3 weightless. The forces and torques acting on 4 are shown in Figure 6.*



Figure 6. Free body diagrams of the members of a mechanism with only link 4 considered to have mass.

Summing the moments about O_B yields:

$$\bar{r}_{g4} \times (-m_4\bar{a}_4) + \bar{r}_4 \times \bar{F}_{34}' - I_4\ddot{\theta}_4 + T_{EX} = 0. \tag{15}$$

This vector equation, when expanded, is seen to contain two unknowns, the components, $F_{34}'^{x}$ and $F_{34}'^{y}$. A second relation in these unknowns is needed. Since link 3 is considered weightless, it may carry no moment and is thus a "two-force" member. That is, forces may only be transmitted along its length. Therefore, the direction of $\bar{F}_{34}'$ is known.

---

* Quantities with bars are vectors; those with x and y are components in these directions. The primes indicate which forces are being considered. One prime, those due to inertia of link 4; two primes, those due to inertia of link 3; and three primes, those due to inertia of link 2.

$$\frac{F_{34}^{\prime y}}{F_{34}^{\prime x}} = \tan\theta_3 . \tag{16}$$

Equations (15) and (16) may be solved simultaneously for $F_{34}^{\prime x}$ and $F_{34}^{\prime y}$.

$$F_{34}^{\prime x} = \frac{-r_{g4}^{x}(-m_4 a_4^{y}) + r_{g4}^{y}(-m_4 a_4^{x}) + I\ddot{\theta}_4 - T_{EX}}{r_4^{x}\tan\theta_3 - r_4^{y}}, \tag{17}$$

$$F_{34}^{\prime y} = F_{34}^{\prime x}\tan\theta_3 . \tag{18}$$

Now, by summing the forces on link 4, the reaction $\bar{F}_{14}^{\prime}$ can be found.

$$F_{14}^{\prime x} = m_4 a_4^{x} - F_{34}^{\prime x}, \tag{19}$$

$$F_{14}^{\prime y} = m_4 a_4^{y} - F_{34}^{\prime y} . \tag{20}$$

Since $\bar{F}_{34}^{\prime}$ has equal and opposite reactions through links 2 and 3,

$$\bar{F}_{32}^{\prime} = \bar{F}_{12}^{\prime} = \bar{F}_{34}^{\prime} . \tag{21}$$

The torque which must be applied to link 2 to balance these forces is

$$\bar{T}_2^{\prime} = -(\bar{r}_2 \times \bar{F}_{32}^{\prime}). \tag{22}$$

A similar procedure is carried out with links 2 and 4 considered weightless and only 3 having mass. Moments are taken about "A".



Figure 7. Free body diagrams of the members of a mechanism with only link 3 considered to have mass.

Knowing the direction of $\bar{F}_{43}''$ the relation

$$\frac{F_{43}''^{y}}{F_{43}''^{x}} = \tan \theta_4$$

is also available, and thus expressions for $F_{43}''^{x}$ and $F_{43}''^{y}$ similar to equations (17) and (18) are obtained. Subsequently, $\bar{F}_{32}''$, $\bar{F}_{21}''$, $\bar{F}_{41}''$, and $\bar{T}_2''$ can be calculated.

Next, links 3 and 4 are considered weightless. If link 2 rotates at constant angular velocity, it produces only a radial bearing reaction.

$$\bar{F}_{21}''' = \bar{r}_{g2} \times (-m_2\bar{a}_{g2}).$$

Finally, all these quantities may be superpositioned, yielding the results:

$$\bar{F}_{21} = \bar{F}'_{21} + \bar{F}''_{21} + F'''_{21}$$

$$\bar{F}_{41} = \bar{F}'_{41} + \bar{F}''_{41}$$

$$\bar{F}_{32} = \bar{F}'_{32} + \bar{F}''_{32}$$

$$\bar{F}_{34} = \bar{F}'_{34} + \bar{F}''_{34}$$

$$\bar{F}_{SHAKE} = \bar{F}_{21} + \bar{F}_{41}$$

$$\bar{T}_{2} = \bar{T}'_{2} + \bar{T}''_{2}.$$

These include the force on each of the bearings, the total force tending to shake the mounting, and the torque required to drive link 2 at a prescribed velocity. [4,8,9]

THE PROGRAM AND ITS USE

Before continuing further, it would be well to list all the assumptions which must be made in order to use the program as it is listed in Appendix C.

1. The mechanism is a crank-rocker.
2. The links are rigid.
3. The friction forces are negligible.
4. Gravity forces are negligible.
5. Link 2 is driven at constant angular velocity.
6. Only link 4 may have external torque. (The driving torque on link 2 is a quantity to be calculated)

The program proceeds in very systematic fashion, exactly in the order in which the analysis proceeded in the foregoing sections. All velocities, accelerations, and forces are carried along as their x and y components. A subroutine is employed to combine these components into a magnitude and direction. Thus each time CALL VECTOR (X, Y, MXY, DXY) appears, 2 components, X and Y, are taken to the subroutine, and 2 quantities, magnitude and direction, are returned.

An effort was made to closely correlate the names used in the program to those which appear in the analysis. Some examples are:

TH2	-	θ_2
TH2D	-	$\dot{\theta}_2$
FP32	-	$\bar{F}'_{32}$
TPP2	-	$\bar{T}''_2$
TEX	-	T_{EX}
W2	-	Weight of link 2
ZM2	-	Mass of link 2 .

On examination of the program, the other quantities should be easily recognized.



Figure 8. A mechanism labeled with quantities as they appear in t he program.

Figure 8 shows a general crank-rocker mechanism with the input parameters labeled as they appear in the program. The data cards must be arranged as follows, according to 1 FORMAT:

W2, W3, W4	- weights of the links (lbs)
ZI2, ZI3, ZI4	- mass moments of inertia about O_A, A, O_B (in-lbs-sec^2)
R1, R2, R3, R4	- lengths of links (in)
ADEL2, ADEL3, ADEL4	- angular position measured + or - from R2, R3, or R4 (degrees)
RG2, RG3, RG4	- distance to center of mass (in)
TH2D	- angular velocity of link 2 (rad/sec)
THZERO, DELTA, THMAX	- initial value of θ_2 increment of θ_2 (degrees) final value of θ_2

As many data cards of the last type as are needed may be put in, one after the other. That is, over some range where great accuracy is required, it may be necessary to consider the mechanism using small increments of θ_2. Over other ranges, large increments of θ_2 may provide sufficiently accurate results. The program will stop when no more cards are available for reading. This arrangement also allows a mechanism to be analyzed at only one position, simply by entering THZERO = THMAX on the last card.

A kinematic analysis alone is achieved by removing all the cards related to forces and torques (statement numbers 133-202), and their corresponding WRITE statements. Accelerations of many points along a link may be obtained by incrementing the length of RG along one of the links.

AN EXAMPLE PROBLEM



Figure 9. Configuration of the example problem.

An example is worked to completely check the program, and also to illustrate how some of the results may be interpreted. The mechanism in Figure 9 is used. Links 2 and 4 are deliberately left unbalanced to make the resulting bearing reactions vary over a large range. The input parameters are as follows:

W2, W3, W4	3.0 8.0 6.0	(lbs)
ZI2, ZI3, ZI4	0* 0.5 0.4	(in-lbs-sec^2)
R1, R2, R3, R4	12.0 4.0 16.0 12.0	(in)
ADEL2, ADEL3, ADEL4	0.0 -15.0 +15.0	(degrees)
RG2, RG3, RG4	2.0 8.0 6.0	(in)
THZD	50.0	(rad/sec)
THZERO, DELTA, THMAX	0 10.0 360.0	(degrees)

* Note that ZI2 may have any value since $\ddot{\theta}_2 = 0$.

Also in the program proper, statement number 74 is made TEX = 0.0. This states that there is no external torque on link 4. In other uses this could be TEX = CONSTANT or TEX = $f(\theta_4)$, where the external torque may depend on the angular position of link 4.

Typical output sheets are shown in Figures 10 and 11. In this case there are 36 such sheets, one for each 10° increment of θ_2. It is immediately seen that in ranges where accelerations are high, the forces change very rapidly in a 10° rotation of link 2. It is advisable to run the program again, this time making the increments of θ_2 smaller in these critical regions. In this instance the program was rerun using 2° increments over the whole range to increase the accuracy in plotting the results.

A preliminary check, which in effect considers all the results, involves the torque vs. θ_2 curve shown in Figure 12. If this plot is correct, in all probability the values of forces, accelerations, and velocities are also correct, as these quantities are all applied when calculating the driving torque.

Since the mechanism is not working against an external torque (TEX = 0) and since the angular velocity of link 2 is constant, the total work done on the system must equal zero.

Therefore:

$$\int_0^{2\pi} \bar{T}_2 \cdot \overline{d\theta}_2 = 0 .$$

Since $\bar{T}_2$ is always in the same direction as θ_2 this reduces to

$$\int_0^{2\pi} T_2 \, d\theta = 0.$$

The graphical integration of the $\overline{T_2}$ vs. θ_2 curve gives a result of almost zero, the error probably resulting from the approximate nature of the process (straight-line approximation between the points spaced at 10°). Notice that if there is an external torque applied to link 4, this integration should equal the work done to overcome this torque.

Assured that all the results are correct, the other output quantities can be plotted and examined. This is done in Figures 13 through 17. Many conclusions of interest in a mechanical design problem can be drawn from these graphs.

Consider the problem of choosing bearings for the joints of a linkage. Of primary interest is the maximum value of the bearing reactions. To find these values by hand computation at a few points would almost be impossible. Figure 13 shows that there can be several "maximums" over a complete cycle of operation. Simply because the value of the force sharply deflects at some point is not conclusive proof that the greatest force on the bearing has been found. From these graphs the maximum value can be immediately obtained.

It is interesting to note that these bearings may be loaded twice at the same place in one cycle of operation. This is particularly evident in Figures 13 and 15 (where the curves cross). Conclusions such as this would be significant when considering the fatigue characteristics of a bearing under cyclic loading.

Figures 13 and 14 show that over certain angular ranges no force is applied to the bearings. These would be ideal positions for the placement of oil grooves or holes, since no stress would ever be applied to these potential trouble spots. Only through a solution which considers the whole range of operation is such information available.

The direction of loading at each of the bearings is also of interest. For example, in Figure 14 the loading relative to link 4 can be seen. It is obvious that the bearing mount would have to sustain much greater loads in a direction along the axis of the link than perpendicular to its axis. Such information may be useful where the weight of the link must be kept at a minimum.

This mechanism was purposely chosen to exhibit large unbalanced forces. The shaking force diagram indicates sizable reactions on the mounting in both the X and Y directions. From Figure 16 it can be seen that the mount must be made stronger in the horizontal direction than in the vertical, since the maximum horizontal component of the shaking force is almost three times as large as the maximum vertical. This program could also be useful in attempting to balance such a mechanism.

The kinematic portion of the solution may also find applications. It may be necessary to find a position in a mechanism which has a particular velocity or acceleration. Such points could easily be found with this program.

THETA 2	THETA 3	THETA 4
40.0	32.1	67.3

ANG VEL 3	ANG VEL 4	ANG ACC 3	ANG ACC 4
-.9955E 01	.3961E 01	.1081E 04	.1636E 04

EX TORQ ON 4
.0000E-99

VEL G2	VEL A	VEL G3	VEL B	VEL G4
.1000E 03	.1999E 03	.1303E 03	.4753E 02	.2376E 02
129.9	130.0	143.7	157.3	172.3

ACC G2	ACC A	ACC G3	ACC B	ACC G4
.5000E 04	.1000E 05	.1108E 05	.1963E 05	.9819E 04
219.9	219.9	171.6	157.9	172.9

F 3 ON 2	F 3 ON 4	F 2 ON 1	F 4 ON 1
.5230E 03	.3540E 03	.5606E 03	.2885E 03
24.8	225.6	25.9	250.5

SHAKE	X SHAKE	Y SHAKE
.4092E 03	.4083E 03	-.2707E 02
356.2		

DRIVING TORQ
.5456E 03

Figure 10. Example problem.
Typical output.

THETA 2	THETA 3	THETA 4
50.0	30.4	68.6

ANG VEL 3	ANG VEL 4	ANG ACC 3	ANG ACC 4
-.6476E 01	.9002E 01	.9120E 03	.1257E 04

EX TORQ ON 4
.0000E-99

VEL G2	VEL A	VEL G3	VEL B	VEL G4
.1000E 03	.1999E 03	.1600E 03	.1080E 03	.5401E 02
140.0	140.0	150.5	158.6	173.6

ACC G2	ACC A	ACC G3	ACC B	ACC G4
.5000E 04	.1000E 05	.8728E 04	.1511E 05	.7558E 04
229.9	229.9	184.7	162.3	177.3

F 3 ON 2	F 3 ON 4	F 2 ON 1	F 4 ON 1
.3844E 03	.2247E 03	.4201E 03	.1637E 03
25.5	222.1	27.7	252.4

SHAKE	X SHAKE	Y SHAKE
.3249E 03	.3225E 03	.3928E 02
6.9		

DRIVING TORQ
.6371E 03

Figure 11. Example problem.
Typical output.



Figure 12. Example problem. Driving torque applied to link 2 versus θ_2



Degree markings around the curve indicate approximate corresponding positions of θ_2.

Figure 13. Example problem.
Polar diagram of force 4 on 1.

Dotted outline indicates
the relative position of
link 4



Degree markings around the
curve indicate approximate
corresponding positions of θ_2

Figure 14. Example problem.
Polar diagram of force 3 on 4,
relative to link 4.



Figure 15. Example problem.
Polar diagram of force 2 on 1.



Figure 16. Example problem.
Horizontal and vertical components of
the shaking force versus θ_z



Figure 17. Example problem.
Polar diagram of total shaking force.

CONCLUSION

The main advantages of this method of solution are, of course, the speed and accuracy with which the results are obtained. The program solves a mechanism in one position for all the unknown parameters with 8 place accuracy in approximately 10 seconds. The importance of having a solution for the complete range of operation of a mechanism has also been emphasized.

This report has presented only a few examples of how the information available from this program may be used in design situations. An attempt was made to show that very real and practical problems can be solved. It has been emphasized that the program was written in a form which closely follows the analysis in the text, making its application to other problems as simple as possible. A wide range of more specific problems which relate to the dynamics of a mechanism could be approached with simple modifications .

ACKNOWLEDGMENT

I express my appreciation to Dr. H. S. Walker of the Department of Mechanical Engineering, Kansas State University, for his guidance and counsel throughout the course of this work as my teacher and advisor.

I am grateful to my wife, Karen, for typing this report and for her patience and understanding during its preparation.

REFERENCES

1. Crossley, F. R. Erskine, ed., Proceedings of the International Conference for Teachers of Mechanisms, Shoe String Press, 1961.

2. Hrones and Nelson, Analysis of the Four Bar Linkage, John Wiley and Son, Inc., New York, 1954.

3. Martin, George H., "Calculating Velocities and Accelerations in Four-Bar Linkages", Machine Design Magazine, April 17, 1958, p. 146.

4. Shigley, Joseph E., Theory of Machines, McGraw-Hill Book Co., Inc., New York, 1961.

5. Hinkle, Rolland T., Kinematics of Machines, Prentice-Hall, Inc., Englewood Cliffs, N. J., 1962.

6. Hartenberg, R. S., "Complex Numbers and Four Bar Linkages", Machine Design Magazine, March 20, 1958, pp. 156-163.

7. Raven, Francis H., "Velocity and Acceleration Analysis of Plane and Space Mechanisms by Means of Independent Position Equations", Journal of Applied Mechanics, March 1958, pp. 1-6.

8. Horschorn, Jeremy, Kinematics and Dynamics of Plane Mechanisms, McGraw-Hill Book Co., Inc., New York, 1962.

9. Maxwell, Robert L., Kinematics and Dynamics of Machinery, Prentice-Hall, Inc., Englewood Cliffs, N. J., 1960.

APPENDIX A

DERIVATION OF GEOMETRY



CASE I



CASE II

θ_2, a, b, c, d ARE KNOWN

CONSTRUCT $O_B A = \ell$

FROM THE LAW OF COSINES:

$$\ell = \sqrt{a^2 + d^2 - 2ad\cos\theta_2}$$

SINCE $\cos\theta = \cos(-\theta)$ THIS APPLIES TO BOTH CASES

$$b^2 = \ell^2 + c^2 - 2\ell c\cos\lambda$$

$$\lambda = \cos^{-1}\left(\frac{\ell^2 + c^2 - b^2}{2\ell c}\right)$$

$$c^2 = \ell^2 + b^2 - 2\ell b\cos\psi$$

$$\psi = \cos^{-1}\left(\frac{\ell^2 + b^2 - c^2}{2\ell b}\right)$$

$$a^2 = \ell^2 + d^2 - 2\ell d \cos\beta$$

$$\beta = \cos^{-1}\left(\frac{\ell^2 + d^2 - a^2}{2\ell d}\right)$$

FROM THE ILLUSTRATIONS:

CASE I $(0^\circ \le \theta_2 \le 180^\circ)$

$$\theta_3 = \psi - \beta$$

$$\theta_4 = 180^\circ - (\lambda + \beta)$$

CASE II $(180^\circ \le \theta_2 \le 360^\circ)$

$$\theta_3 = \psi + \beta$$

$$\theta_4 = 180^\circ - (\lambda - \beta)$$

APPENDIX B

DERIVATION OF ANGULAR VELOCITY AND ACCELERATION



$\theta_1 = 0$
$\theta_2 = $ KNOWN
$\dot{\theta}_2 = $ KNOWN
$\ddot{\theta}_2 = $ KNOWN $= 0$

$$r_2 e^{i\theta_2} + r_3 e^{i\theta_3} = r_1 e^{i\theta_1} + r_4 e^{i\theta_4}$$

DIFFERENTIATING

$$i r_2 \dot{\theta}_2 e^{i\theta_2} + i r_3 \dot{\theta}_3 e^{i\theta_3} = i r_4 \dot{\theta}_4 e^{i\theta_4}$$

$$i r_2 \dot{\theta}_2 (\cos\theta_2 + i \sin\theta_2) + i r_3 \dot{\theta}_3 (\cos\theta_3 + i \sin\theta_3) = i r_4 \dot{\theta}_4 (\cos\theta_4 + i \sin\theta_4)$$

IMAGINARY –

$$r_2 \dot{\theta}_2 \cos\theta_2 + r_3 \dot{\theta}_3 \cos\theta_3 = r_4 \dot{\theta}_4 \cos\theta_4$$

REAL –

$$r_2 \dot{\theta}_2 \sin\theta_2 + r_3 \dot{\theta}_3 \sin\theta_3 = r_4 \dot{\theta}_4 \sin\theta_4$$

$$-r_3 \cos\theta_3 (\dot{\theta}_3) + r_4 \cos\theta_4 (\dot{\theta}_4) = r_2 \dot{\theta}_2 \cos\theta_2$$
$$-r_3 \sin\theta_3 (\dot{\theta}_3) + r_4 \sin\theta_4 (\dot{\theta}_4) = r_2 \dot{\theta}_2 \sin\theta_2$$

SOLVING FOR $\dot{\theta}_3$ AND $\dot{\theta}_4$

$$\dot{\theta}_3 = \frac{\begin{vmatrix} r_2 \dot{\theta}_2 \cos\theta_2 & r_4 \cos\theta_4 \\ r_2 \dot{\theta}_2 \sin\theta_2 & r_4 \sin\theta_4 \end{vmatrix}}{\begin{vmatrix} -r_3 \cos\theta_3 & r_4 \cos\theta_4 \\ -r_3 \sin\theta_3 & r_4 \sin\theta_4 \end{vmatrix}}$$

$$\dot{\theta}_3 = \frac{r_2 r_4 \dot{\theta}_2 (\cos\theta_2 \sin\theta_4 - \sin\theta_2 \cos\theta_4)}{r_3 r_4 (-\cos\theta_3 \sin\theta_4 + \sin\theta_3 \cos\theta_4)}$$

$$\dot{\theta}_3 = \frac{r_2}{r_3} \dot{\theta}_2 \frac{\sin(\theta_4 - \theta_2)}{\sin(\theta_3 - \theta_4)}$$

$$\dot{\theta}_4 = \frac{\begin{vmatrix} -r_3 \cos\theta_3 & r_2\dot{\theta}_2 \cos\theta_2 \\ -r_3 \sin\theta_3 & r_2\dot{\theta}_2 \sin\theta_2 \end{vmatrix}}{\begin{vmatrix} -r_3 \cos\theta_3 & r_4 \cos\theta_4 \\ -r_3 \sin\theta_3 & r_4 \sin\theta_4 \end{vmatrix}}$$

$$\dot{\theta}_4 = \frac{r_3 r_2 \dot{\theta}_2 (\sin\theta_3 \cos\theta_2 - \cos\theta_3 \sin\theta_2)}{r_3 r_4 \sin(\theta_3 - \theta_4)}$$

$$\dot{\theta}_4 = \frac{r_2}{r_4} \dot{\theta}_2 \frac{\sin(\theta_3 - \theta_2)}{\sin(\theta_3 - \theta_4)}$$

DIFFERENTIATING THE ORIGINAL EXPRESSION TWICE

$$i r_2 \ddot{\theta}_2 e^{i\theta_2} - r_2 \dot{\theta}_2^2 e^{i\theta_2} + i r_3 \ddot{\theta}_3 e^{i\theta_3} - r_3 \dot{\theta}_3^2 e^{i\theta_3} = i r_4 \ddot{\theta}_4 e^{i\theta_4} - r_4 \dot{\theta}_4^2 e^{i\theta_4}$$

IMAGINARY

$$r_2 \ddot{\theta}_2 \cos\theta_2 - r_2 \dot{\theta}_2^2 \sin\theta_2 + r_3 \ddot{\theta}_3 \cos\theta_3 - r_3 \dot{\theta}_3^2 \sin\theta_3 = r_4 \ddot{\theta}_4 \cos\theta_4 - r_4 \dot{\theta}_4^2 \sin\theta_4$$

REAL

$$r_2 \ddot{\theta}_2 \sin\theta_2 + r_2 \dot{\theta}_2^2 \cos\theta_2 + r_3 \ddot{\theta}_3 \sin\theta_3 + r_3 \dot{\theta}_3^2 \cos\theta_3 = r_4 \ddot{\theta}_4 \sin\theta_4 + r_4 \dot{\theta}_4^2 \cos\theta_4$$

$$-r_3\cos\theta_3(\ddot{\theta}_3) + r_4\cos\theta_4(\ddot{\theta}_4) = r_2\ddot{\theta}_2\cos\theta_2 - r_2\dot{\theta}_2^2\sin\theta_2 - r_3\dot{\theta}_3^2\sin\theta_3 + r_4\dot{\theta}_4^2\sin\theta_4$$

$$-r_3\sin\theta_3(\ddot{\theta}_3) + r_4\sin\theta_4(\ddot{\theta}_4) = r_2\ddot{\theta}_2\sin\theta_2 + r_2\dot{\theta}_2^2\cos\theta_2 + r_3\dot{\theta}_3^2\cos\theta_3 - r_4\dot{\theta}_4^2\cos\theta_4$$

THE RIGHT SIDE OF EACH EQUATION CONTAINS ONLY KNOWN QUANTITIES (NOTICE THAT $\ddot{\theta}_2 = 0$)

SOLVING FOR $\ddot{\theta}_3$ AND $\ddot{\theta}_4$

$$\ddot{\theta}_3 = \frac{-r_2\cancelto{0}{\ddot{\theta}_2}\sin(\theta_4-\theta_2) + r_2\dot{\theta}_2^2\cos(\theta_4-\theta_2) + r_3\dot{\theta}_3^2\cos(\theta_4-\theta_3) - r_4\dot{\theta}_4^2}{r_3\sin(\theta_4-\theta_3)}$$

$$\ddot{\theta}_4 = \frac{r_2\cancelto{0}{\ddot{\theta}_2}\sin(\theta_3-\theta_2) - r_2\dot{\theta}_2^2\cos(\theta_3-\theta_2) + r_4\dot{\theta}_4^2\cos(\theta_3-\theta_4) - r_3\dot{\theta}_3^2}{r_4\sin(\theta_3-\theta_4)}$$

APPENDIX C

THE PROGRAM

```
00001 FORMAT(11X,3F12.1/)
00002 FORMAT(15X,4E12.4/)
00003 FORMAT(11X,4F12.1/)
00004 FORMAT(11X,1F12.1/)
00005 FORMAT(15X,1E12.4/)
00006 FORMAT(15X,3E12.4/)
00007 FORMAT(/)
00008 FORMAT(11X,5F12.1/)
00009 FORMAT(15X,5E12.4/)
00010 FORMAT(//18X,7HTHETA 2,5X,7HTHETA 3,5X,7HTHETA 4/)
00011 FORMAT(//18X,9HANG VEL 3,3X,9HANG VEL 4,3X,
     19HANG ACC 3,3X,9HANG ACC 4/)
00012 FORMAT(//18X,12HEX TORQ ON 4/)
00013 FORMAT(//18X,6HVEL G2,6X,5HVEL A,7X,6HVEL G3,6X,
     15HVEL B,7X,6HVEL G4/)
00014 FORMAT(//18X,6HACC G2,6X,5HACC A,7X,6HACC G3,6X,
     15HACC B,7X,6HACC G4/)
00015 FORMAT(//18X,8HF 3 ON 2,4X,8HF 3 ON 4,4X,
     18HF 2 ON 1,4X,8HF 4 ON 1/)
00016 FORMAT(//18X,5HSHAKE,7X,7HX SHAKE,5X,7HY SHAKE/)
00017 FORMAT(//18X,12HDRIVING TORQ/)
00018 FORMAT(1H1)
00019 FORMAT(4F8.4)
00020 FORMAT(37H THIS IS NOT A CRANK ROCKER MECHANISM)
      READ(1,19)W2,W3,W4
      READ(1,19)ZI2,ZI3,ZI4
      READ(1,19)R1,R2,R3,R4
      READ(1,19)ADEL2,ADEL3,ADEL4
      READ(1,19)RG2,RG3,RG4
      READ(1,19)TH2D
00027 READ(1,19)THZERO,DELTA,THMAX
      G=386.0880
      ZM2=W2/G
      ZM3=W3/G
      ZM4=W4/G
      TH=THZERO-DELTA
      I=0
      RAD=57.295779
      PI=3.14159265
      SQ2D=TH2D*TH2D
      DEL2=ADEL2/RAD
      DEL3=ADEL3/RAD
      DEL4=ADEL4/RAD
      A=R3/R2
      B=R4/R2
      C=R1/R2
      IF(A.GT.1.0)GOTO45
      GOTO52
00045 IF(B.GT.1.0)GOTO47
      GOTO52
00047 IF(C.GT.1.0)GOTO49
      GOTO52
00049 IF(C.LT.(A+B-1.0))GOTO51
```

```
      GOTO52
00051 IF(C.GT.(ABS(B-A)+1.0))GOTO54
00052 WRITE(3,20)
      GOTO237
00054 I=I+1
      ZZZ=I
      ATH2=TH+ZZZ*DELTA
      TH2=ATH2/RAD
      CTH2=COS(TH2)
      ZL=SQRT(R2*R2+R1*R1-2.0*R2*R1*CTH2)
      CPSI=(-(R4*R4)+ZL*ZL+R3*R3)/(2.0*ZL*R3)
      PSI=ARCCOS(CPSI)
      CZLAM=(ZL*ZL+R4*R4-R3*R3)/(2.0*ZL*R4)
      ZLAM=ARCCOS(CZLAM)
      CBETA=(ZL*ZL+R1*R1-R2*R2)/(2.0*ZL*R1)
      BETA=ARCCOS(CBETA)
      IF(TH2.GE.PI)GOTO70
      TH3=PSI-BETA
      TH4=PI-ZLAM-BETA
      GOTO72
00070 TH3=PSI+BETA
      TH4=PI-ZLAM+BETA
00072 ATH3=TH3*RAD
      ATH4=TH4*RAD
00074 TEX=0.0
      S42=SIN(TH4-TH2)
      S34=SIN(TH3-TH4)
      S32=SIN(TH3-TH2)
      C42=COS(TH4-TH2)
      C43=COS(TH4-TH3)
      S43=SIN(TH4-TH3)
      C32=COS(TH3-TH2)
      C34=COS(TH3-TH4)
      STHDE2=SIN(TH2+DEL2)
      CTHDE2=COS(TH2+DEL2)
      STHDE3=SIN(TH3+DEL3)
      CTHDE3=COS(TH3+DEL3)
      STHDE4=SIN(TH4+DEL4)
      CTHDE4=COS(TH4+DEL4)
      STH4=SIN(TH4)
      CTH4=COS(TH4)
      STH3=SIN(TH3)
      CTH3=COS(TH3)
      STH2=SIN(TH2)
      CTH2=COS(TH2)
      TH3D=(R2/R3)*TH2D*(S42/S34)
      TH4D=(R2/R4)*TH2D*(S32/S34)
      SQ3D=TH3D*TH3D
      SQ4D=TH4D*TH4D
      TH3DD=(R2*SQ2D*C42+R3*SQ3D*C43-R4*SQ4D)/(R3*S43)
      TH4DD=(-R2*SQ2D*C32+R4*SQ4D*C34-R3*SQ3D)/(R4*S34)
      R2X=R2*CTH2
      R2Y=R2*STH2
```

```
      RGX=RG2*CTHDE2
      RGY=RG2*STHDE2
      R3X=R3*CTH3
      R3Y=R3*STH3
      RG3X=RG3*CTHDE3
      RG3Y=RG3*STHDE3
      R4X=R4*CTH4
      R4Y=R4*STH4
      RG4X=RG4*CTHDE4
      RG4Y=RG4*STHDE4
      VAX=-R2*TH2D*STH2
      VAY=R2*TH2D*CTH2
      AAX=-R2*SQ2D*CTH2
      AAY=-R2*SQ2D*STH2
      VG2X=-RG2*TH2D*STHDE2
      VG2Y=RG2*TH2D*CTHDE2
      AG2X=-RG2*SQ2D*CTHDE2
      AG2Y=-RG2*SQ2D*STHDE2
      VBX=-R4*TH4D*STH4
      VBY=R4*TH4D*CTH4
      ABX=-R4*SQ4D*CTH4-R4*TH4DD*STH4
      ABY=-R4*SQ4D*STH4+R4*TH4DD*CTH4
      VG4X=-RG4*TH4D*STHDE4
      VG4Y=RG4*TH4D*CTHDE4
      AG4X=-RG4*SQ4D*CTHDE4-RG4*TH4DD*STHDE4
      AG4Y=-RG4*SQ4D*STHDE4+RG4*TH4DD*CTHDE4
      VG3X=+VAX-RG3*TH3D*STHDE3
      VG3Y=VAY+RG3*TH3D*CTHDE3
      AG3X=AAX-RG3*SQ3D*CTHDE3-RG3*TH3DD*STHDE3
      AG3Y=AAY-RG3*SQ3D*STHDE3+RG3*TH3DD*CTHDE3
00133 FG4X=-ZM4*AG4X
      FG4Y=-ZM4*AG4Y
      TI4=-ZI4*TH4DD
      IF(CTH3.EQ.0.0)GOTO139
      TAN3=STH3/CTH3
      GOTO140
00139 TAN3=0.9E+25
00140 FP34X=(-RG4X*FG4Y+RG4Y*FG4X-TI4-TEX)/(R4X*TAN3-R4Y)
      FP43X=-FP34X
      FP34Y=FP34X*TAN3
      FP43Y=-FP34Y
      FP14X=-FG4X-FP34X
      FP41X=-FP14X
      FP14Y=-FG4Y-FP34Y
      FP41Y=-FP14Y
      FP23X=FP34X
      FP32X=-FP23X
      FP23Y=FP34Y
      FP32Y=-FP23Y
      FP12Y=FP23Y
      FP21Y=-FP12Y
      FP12X=FP23X
      FP21X=-FP12X
```

```
      TP2=-R2X*FP32Y+R2Y*FP32X
      TI3=-ZI3*TH3DD
      FG3X=-ZM3*AG3X
      FG3Y=-ZM3*AG3Y
      IF(CTH4.EQ.0.0)GOTO163
      TAN4=STH4/CTH4
      GOTO164
00163 TAN4=0.9E+25
00164 FPP43X=(-RG3X*FG3Y+RG3Y*FG3X-TI3)/(R3X*TAN4-R3Y)
      FPP34X=-FPP43X
      FPP43Y=FPP43X*TAN4
      FPP34Y=-FPP43Y
      FPP23X=-FPP43X-FG3X
      FPP32X=-FPP23X
      FPP23Y=-FPP43Y-FG3Y
      FPP32Y=-FPP23Y
      FPP14X=FPP43X
      FPP41X=-FPP14X
      FPP14Y=FPP43Y
      FPP41Y=-FPP14Y
      FPP12X=FPP23X
      FPP21X=-FPP12X
      FPP12Y=FPP23Y
      FPP21Y=-FPP12Y
      TPP2=-R2X*FPP32Y+R2Y*FPP32X
      FG2X=-ZM2*AG2X
      FG2Y=-ZM2*AG2Y
      F3P12X=-FG2X
      F3P21X=-F3P12X
      F3P12Y=-FG2Y
      F3P21Y=-F3P12Y
      FA32X=FP32X+FPP32X
      FA32Y=FP32Y+FPP32Y
      FB34X=FP34X+FPP34X
      FB34Y=FP34Y+FPP34Y
      TORQ=TP2+TPP2
      FOAX=FP21X+FPP21X+F3P21X
      FOAY=FP21Y+FPP21Y+F3P21Y
      FOBX=FP41X+FPP41X
      FOBY=FP41Y+FPP41Y
      SHAKX=FOAX+FOBX
      SHAKY=FOAY+FOBY
      CALLVECTOR(FA32X,FA32Y,FA32,DFA32)
      CALLVECTOR(FB34X,FB34Y,FB34,DFB34)
      CALLVECTOR(FOAX,FOAY,FOA,DFOA)
      CALLVECTOR(FOBX,FOBY,FOB,DFOB)
00202 CALLVECTOR(SHAKX,SHAKY,SHAK,DSHAK)
      CALLVECTOR(VAX,VAY,VA,DVA)
      CALLVECTOR(AAX,AAY,AA,DAA)
      CALLVECTOR(VG2X,VG2Y,VG2,DVG2)
      CALLVECTOR(AG2X,AG2Y,AG2,DAG2)
      CALLVECTOR(VBX,VBY,VB,DVB)
      CALLVECTOR(ABX,ABY,AB,DAB)
```

```
      CALLVECTOR(VG4X,VG4Y,VG4,DVG4)
      CALLVECTOR(AG4X,AG4Y,AG4,DAG4)
      CALLVECTOR(VG3X,VG3Y,VG3,DVG3)
      CALLVECTOR(AG3X,AG3Y,AG3,DAG3)
      WRITE(3,7)
      WRITE(3,10)
      WRITE(3,1)ATH2,ATH3,ATH4
      WRITE(3,11)
      WRITE(3,2)TH3D,TH4D,TH3DD,TH4DD
      WRITE(3,12)
      WRITE(3,5)TEX
      WRITE(3,13)
      WRITE(3,9)VG2,VA,VG3,VB,VG4
      WRITE(3,8)DVG2,DVA,DVG3,DVB,DVG4
      WRITE(3,14)
      WRITE(3,9)AG2,AA,AG3,AB,AG4
      WRITE(3,8)DAG2,DAA,DAG3,DAB,DAG4
      WRITE(3,15)
      WRITE(3,2)FA32,FB34,FOA,FOB
      WRITE(3,3)DFA32,DFB34,DFOA,DFOB
      WRITE(3,16)
      WRITE(3,6)SHAK,SHAKX,SHAKY
      WRITE(3,4)DSHAK
      WRITE(3,17)
      WRITE(3,5)TORQ
      WRITE(3,18)
      IF(ATH2.GT.THMAX)GOTO27
      GOTO54
00237 CONTINUE
      STOP
      END
```

```
      FUNCTION ARCCOS(X)
      PI=3.14159265358979323846
      ZMULT=1.0
      PROD=1.0
      SAVE=0.1E-50
      XX=X
      X2=XX*XX
      IF(ABS(XX).LE.0.7) GO TO 5
      XX=SQRT(1.0-X2)
      X2=XX*XX
    5 SUM=(PI/2.0)-XX
      VAL=XX
      A=0.0
  100 A=A+2.0
      PROD=PROD*A
      VAL=VAL*X2
      ZMULT=ZMULT*(A-1.0)
      SUM=SUM-(ZMULT*VAL)/(PROD*(A+1.0))
      IF(ABS(SUM-SAVE).LT.0.00000001) GO TO 200
      SAVE=SUM
      GO TO 100
  200 IF(X.LT.-0.70) GO TO 103
      IF(X.GT.0.70) GO TO 101
      ARCCOS=SUM
      GO TO 102
  101 ARCCOS=(PI/2.0)-SUM
      GO TO 102
  103 ARCCOS=(PI/2.0)+SUM
  102 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE VECTOR(X,Y,VECT,DIRECT)
      ANG=0.0
      PI=3.14159265
      RAD=57.2957795
      IF(X)36,37,36
   36 ARC=ATAN(ABS(Y)/ABS(X))
      GO TO 41
   37 IF(Y) 38,42,40
   41 IF(X) 30,42,31
   31 IF(Y) 32,33,33
   30 IF(Y) 34,35,35
   32 ANG=2.0*PI-ARC
      GO TO 42
   33 ANG=ARC
      GO TO 42
   34 ANG=PI+ARC
      GO TO 42
   35 ANG=PI-ARC
      GO TO 42
   38 ANG=(3.0*PI)/2.0
      GO TO 42
   40 ANG=PI/2.0
   42 DIRECT=ANG*RAD
      VECT=SQRT(X*X+Y*Y)
      RETURN
      END
```